

6.4 フィルター State Space Model の基礎

工事中

Univariate UC Model

プログラム

```
new; cls;
load data[91,2]=D:\datafile17.txt;
data=(data[:,2]);
data=ln(data);

library maxlik,pgraph;
maxset;
start={3,3,3,1.5,-1};
_max_Algorithm=2;
_max_GradTol=1e-4;
{x,f,g,cov,retcode}=maxprt(maxlik(data,0,&ll,start));
b=ucfilter(x,data);
pqgwin auto;
t=rows(data);
ini=round(0.3*rows(data));
xy(seqa(1,1,t-ini+1),b[ini:t,1]~data[ini:t]);
xy(seqa(1,1,t-ini+1),b[ini:t,2]~zeros(t-ini+1,1));

proc ll(par,data);
  local t,vt,et,wt,phi1,phi2,F,H,Q,R,beta,P,i,ft,eta,lnL,P1,beta1;
  t=rows(data);
  par[1:3]=exp(-par[1:3])/10;
  par[4]=par[4]/(1+abs(par[4]))+par[5]/(1+abs(par[5]));
  par[5]=-par[4]/(1+abs(par[4]))*par[5]/(1+abs(par[5]));
  vt=par[1]; et=par[2]; wt=par[3]; phi1=par[4]; phi2=par[5];
  F=(1~0~0~1) |
    (0~phi1~phi2~0) |
    (0~1~0~0) |
    (0~0~0~1);
  H=(1~1~0~0);
  Q=((vt^2)~0~0~0) |
```

```

        (0~(et^2)~0~0) |
        (0~0~0~0) |
        (0~0~0~(wt^2));
R=0;
beta=zeros(4,1);
P=eye(4)*100;
lnL=zeros(t,1);
i=1;
do while i<=t;
    beta1=F*beta;
    P1=F*P*F'+Q;
    eta=data[i]-H*beta1;
    ft=H*P1*H'+R;
    lnL[i]=-0.5*ln(2*pi*ft)-0.5*eta'inv(ft)*eta;
    beta=beta1+P1*H'inv(ft)*eta;
    P=P1-P1*H'inv(ft)*H*P1;
    i=i+1;
end;
retp(lnL[round(0.1*t):t]);
endp;

proc ucfilter(par,data);
    local t,b,vt,et,wt,phi1,phi2,F,H,Q,R,beta,p,i,ft,eta,P1,beta1;
    t=rows(data);
    b=zeros(t,3);
    vt=par[1]; et=par[2]; wt=par[3]; phi1=par[4]; phi2=par[5];
    F=(1~0~0~1) |
        (0~phi1~phi2~0) |
        (0~1~0~0) |
        (0~0~0~1);
    H=(1~1~0~0);
    Q=((vt^2)~0~0~0) |
        (0~(et^2)~0~0) |
        (0~0~0~0) |
        (0~0~0~(wt^2));
    R=0;


```

```

beta=zeros(4,1);
P=eye(4)*100;
i=1;
do while i<=t;
    beta1=F*beta;
    P1=F*P*F'+Q;
    eta=data[i]-H*beta1;
    ft=H*P1*H'+R;
    beta=beta1+P1*H'inv(ft)*eta;
    P=P1-P1*H'inv(ft)*H*P1;
    b[i,]=beta[1]~beta[2]~beta[4];
    i=i+1;
end;
retp(b);
endp;

```

グラフ表示（データ（実線）とトレンド（破線））



グラフ表示 (サイクル)

